# C2M14

## Animation of Taylor Polynomials

From our text we have seen that a function $f(x)$ may be written as the sum of an $n^{th}$ degree polynomial $T_n(x)$ and a remainder term $R_n(x)$ which tends to 0 as $n$ approaches infinity. For an open interval $|x-a|<R$

$$f(x)=T_n(x)+R_n(x) \qquad T_n(x)=\sum_{i=0}^{n}\frac{f^{(i)}(a)}{i!}(x-a)^i \qquad \lim_{n\to\infty} R_n(x)=0$$

It should be obvious to the reader at this point that $T_n$ approximates $f$ on the interval $|x-a|<R$ and that as $n$ increases and we take more terms then the approximation improves. It is our objective here to provide a graphical illustration of this process. The function that we will use as our introductory example is $f(x)=\dfrac{1}{1+x}$. Our experience with geometric series allows us to "play Jeopardy" and pose "$\frac{1}{1+x}$ is the sum of what series?" Our response is "What is the sum of $1-x+x^2-x^3+x^4-x^5\ldots$ " This allows us to easily identify $T_3(x)=1-x+x^2-x^3$ as a simple approximation to $f(x)$. Now we will set all this up in Maple and display the functions.

```
> restart:      with(plots):    with(student):
> f:=x->1/(1+x);
```

$$f := x \to \frac{1}{1+x}$$

We are going to set up values to serve as domain and range of our plots and an expression which is really $T_3(x)$.

```
> a:=-2:    b:=2:    c:=-5:    d:=10:
> P:=1-x+x^2-x^3;
```

$$P := 1-x+x^2-x^3$$

```
> plot([f(x),P], x=a..b,y=c..d,color=[blue,red],thickness=[1,2]);
```




Now we begin the process of setting up our sequence of Taylor Polynomials, which we do by identifying the degrees that we want to see displayed. We will select the first 29 Taylor Polynomials. So, we will start with 1 and increase by 1 until we reach 29. Continuing our worksheet,

```
> nstart:=1:    skip:=1:    frameno:=29:
> framenumbers:=[seq(nstart + skip*i, i=0..frameno-1)]:
```

$$framenumbers := [1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,21,22,23,24,25,26,27,28,29]$$

Now we will set up our *frameno* (29) Taylor series, convert them to polynomials, and set up the plots of these polynomials. They will appear in red.

```
> A:=display(seq(plot(convert(taylor(f(x),x=0,i),polynom),
   x=a..b,y=c..d,style=line,thickness=2,numpoints=100,
   title="degree = ".i),i=framenumbers),insequence=true):
```

We need matching plots of $f(x)$, which will appear in blue.

```
> B:=animate(f(x),x=a..b,y=c..d,frames=frameno,color=blue):
```

Combine all this into one plot.

```
> display(A,B,view=[a..b,c..d]);
```

degree = 1




You should see the first Taylor Polynomial, which is $y = 1$, displayed with the graph of $f(x)$. Click on the display and a box should appear around the display. Also, a new menu bar appears so that you have what looks like the buttons for a tape player on the bottom line. The button on the left is "stop" and the button next to it is "play". Click on this button.




You should see the rapid display of the Taylor Polynomials with $f(x)$. Now try the next button to the right, and click on it repeatedly.

If you just want to display a few frames of the animation, it is simpler to just identify the numbers of the frames. The lines below could replace the commands above to get a list of frame numbers.

```
> framenumbers:= [1,3,5,7,9,11,13,15,17,19,21,23,25,27,29];
```

$$framenumbers := [1, 3, 5, 7, 9, 11, 13, 15, 17, 19, 21, 23, 25, 27, 29]$$

```
> frameno:=nops(framenumbers);
```

$$frameno := 15$$

The Maple command `nops` returns the number of elements in the list.

**C3M14 Problems**: 1. Use $-10 \le x \le 10$ and $-10 \le y \le 10$ to set up an animation for the Taylor Polynomials of $\sin(x)$ with $a = 0$.

2. Set up a Taylor Polynomial animation example using $f(x) = \arctan(x)$.

3. Set up a Taylor Polynomial animation example using $f(x) = e^x$. Remember, $e^x$ is `exp(x)` in Maple.